
1 RISC-V Instructions
RISC-V is an assembly language composed of simple instructions that each perform a single task
such as addition of two numbers or storing data to memory. Below is a comparison between RISC-
V code and its equivalent C code:

// x in s0, &y in s1  
addi s0, x0, 5 int x = 5;
sw s0, 0(s1) y[0] = x;
mul t0, s0, s0
sw t0, 4(s1) y[1] = x * x;

For your reference, here are some of the basic instructions for arithmetic/bitwise operations and
memory access, which can also be found on the 61C Reference Card.

The below are abbreviations that will be used in the table:
• rs1: Argument register 1
• rs2: Argument register 2
• rd: Destination register
• imm: Immediate value (integer literal constant)
• R[register]: Value contained in register
• inst: One of the instructions in the table

Register-to-register operations (R-type): inst rd rs1 rs2
add Adds R[rs1] and R[rs2] and stores the result in rd
xor Exclusive ORs R[rs1] and R[rs2] and stores the result in rd
mul Multiplies R[rs1] by R[rs2] and stores the result in rd
sll Logical left shifts R[rs1] by R[rs2] and stores the result in rd
srl Logical right shifts R[rs1] by R[rs2] and stores the result in rd
sra Arithmetic right shifts R[rs1] by R[rs2] and stores the result in rd

slt(u) If R[rs1] < R[rs2], puts 1 in rd, else puts 0 (u compares unsigned)

Memory operations
sw rs2 imm(rs1) Stores R[rs2] to the address R[rs1] + imm in memory
lw rd imm(rs1) Loads address R[rs1] + imm from memory into rs2

Branch operations (B-type): inst rs1 rs2 label
bne If rs1 != rs2, jump to label

Branch operations (B-type): inst rs1 rs2 label
beq If rs1 == rs2, jump to label

Jump operations (J-type): inst rd label
jal Stores the next instruction’s address into rd and jumps to label

A RISC-V “immediate” is any numeric constant. For example, addi t0, t0, 20, sw a4, -8(sp),
and lw a1, 0x44(t2) have immediates 20, -8, and 0x44 respectively. Note that there is a limit
to the size (number of bits) of an immediate in any given instruction (depends on what type of
instruction, more on this soon!).

You may also see that there is an “i” at the end of certain instructions, such as addi, slli, etc. This
means that rs2 becomes an “immediate” or an integer instead of a register. There are immediates
in instructions which use an offset such as sw and lw. When coding in RISC-V, always use the
61C reference card for the details of each instruction (the reference card is your friend)!

2 Calling Conventions Pre-Check
2.1 After calling a function and having that function return, the t registers may have been changed

during the execution of the function, while a registers cannot.

2.2 In order to use the saved registers (s0-s11) in a function, we must store their values before using
them and restore their values before returning.

2.3 The stack should only be manipulated at the beginning and end of functions, where the callee-
saved registers are temporarily saved.

3 Calling Conventions
Let’s review what special meaning we assign to each type of register in RISC-V.

Register Convention Saver
x0 Stores zero N/A
sp Stores the stack pointer Callee
ra Stores the return address Caller

a0 - a7 Stores arguments and return values Caller
t0 - t6 Stores temporary values that do not persist after function calls Caller

s0 - s11 Stores saved values that persist after function calls Callee

To save and recall values in registers, we use the sw and lw instructions to save and load words
to and from memory, and we typically organize our functions as follows:

# Prologue  
addi sp, sp, -8 # Room for two registers. (Why?)
sw s0, 0(sp) # Save s0 (or any saved register)
sw s1, 4(sp) # Save s1 (or any saved register)
  
# Code ommitted
  
# Epilogue  
lw s0, 0(sp) # Load s0 (or any saved register)
lw s1, 4(sp) # Load s1 (or any saved register)
addi sp, sp, 8 # Restore the stack pointer

4 Calling Conventions in Code Example
Below is an example of calling conventions in a RISC-V function.

The callee-saved registers (like s0) are saved at the start of the function and restored before
returning, as these registers must be preserved by the function.

The caller-saved registers (like t1 and ra) are saved by the caller before invoking another function,
as the callee can modify these registers. Note: Although ra is a caller-saved register, it is usually
saved at the very beginning and end of the function by convention, as shown below.

func_a: 
   # Prologue: Save callee-saved registers & the return address
   addi sp, sp, -8 # Allocate stack space
   sw ra, 0(sp) # Save return address
   sw s0, 4(sp) # Save s0
  
   addi t1, x0, 10 # Modify t1
   addi s0, x0, 20 # Modify s0
  
   # Save caller-saved registers before function call 
   addi sp, sp, -4 # Allocate more stack space
   sw t1, 0(sp) # Save t1 (caller-saved register)
  
   jal func_b # Call another function
  
   # Restore caller-saved registers after function call
   lw t1, 0(sp) # Restore t1 (caller-saved register)
   addi sp, sp, 4 # Deallocate space for caller-saved register
  
   addi t1, t1, 5 # Modify t1
   addi s0, s0, 5 # Modify s0
  
   # Epilogue: Restore callee-saved registers
   lw ra, 0(sp) # Restore return address
   lw s0, 4(sp) # Restore s0
   addi sp, sp, 8 # Deallocate stack space
  
   ret # Return from func_a

